
Tux Changelog
    


  
    
  
  v0.X.Y - Current


This version is not yet published to hex.pm.
	Add new tux_new installer
	Add test for raising Tux.Error exceptions


  
    
  
  v0.4.0 – June 2024


	Include dispatcher preloads in the final command preloads.
	Refactor and improve Tux.Dispatcher.
	Rename Tux.Error :exit_code to :exitcode.
	Rename Tux.Explainable protocol to Tux.Alertable.
	Replace demo example with recon.
	Rename Tux.Quick module macro to program.
	Remove :warn type from the Tux.Result.
	Add Tux.Exit module for controlling exits.
	Move all errors in the Tux.Errors module.
	Remove Tux.Commands.NotImplemented.
	Show both choices when using Prompt.for_confirmation.
	Emit ConfigReadError and ConfigWriteError on bang config functions.
	Add new dispatcher option :exitwith with possible values :halt or :stop.
	Exit with non-zero status when command modules return errors.


  
    
  
  v0.3.5 – February 2024


	Update pre macro to accept a single command.
	Add compile-time checks for conflicting command names.
	Add cmd support for prefixed commands via sigil_p.


  
    
  
  v0.3.0 – January 2024


	Update the cmd macro to allow registration of multiple names for a command:
e.g. cmd ~w(name another_name), CmdModule.
	Add support for cumulative preloads when using nested dispatchers,
by merging preloads down the command hierarchy.
	Rename the command module callback from info/0 to about/0
	Add Tux.Commands.NotImplemented, which can be used as a
placeholder command module when one hasn't been implemented yet.
	Add the pre/2 macro to simplify the registration of commands
with common preloads.


  
    
  
  v0.2.0 – December 2023


	Add Tux.Init, Tux.Exec & Tux.Show as separate stages in a command life cycle
	Add Tux.Env struct for storing preloads returns and command context
	Add Tux.Colors for writing colored output
	Add Tux.Prompt for user prompting (ints, floats, strings) and confirmations
	Add Tux.Config for reading and writing simple key = value config files
	Add Tux.Explainable protocol as a source for alerts
	Add Tux.Alert for displaying warnings and errors
	Add Tux.Error for creating displayable errors
	Add Tux.Help for creating help messages
	Add Tux.Result for validating returns of commands
	Add Tux.Command for creating command modules
	Add Tux.Dispatcher for creating dispatcher modules and registering
command modules with cmd/3 macro.


  
    
  
  v0.1.0


	Initial commit



  

    
Tux 
    



      
Terminal User Experience
tuxpkg.dev ›
github.com ›
Tux is a modular, dependency-free Elixir library
designed for the speedy creation of elegant command line interfaces
which subscribe to the philosophy "One module per command".
Its modular structure ensures its core functionalities can be overwritten
and composed to fulfill custom needs, and having no dependencies means
it can achieve a higher level of security by minimizing trusted parties.
NOTE: To bootstrap a new tux-based CLI Elixir project, use the
tux_new generator available
on Hex:
$ mix do archive.install hex tux_new + tux.new myapp


  
    
  
  Features


Developing with the tux library involves typically the use of
the following featured modules:
	Tux.Dispatcher - command registration and dispatch macros
	Tux.Command    - command module behaviour
	Tux.Result     - shapes for the command results
	Tux.Prompt     - basic user prompting and answer parsing
	Tux.Error      - user friendly error messages
	Tux.Config     - reading and writing of simple configuration files
	Tux.Help       - command help message construction
	Tux.Case       - command testing macros


  
    
  
  Installation


Add tux to your list of dependencies in mix.exs and optionally update
your .formatter.exs:
# mix.exs
{:tux, "~> 0.4.0"}

# .formatter.exs
import_deps: [:tux]

  
    
  
  Examples


Here's a very short example to illustrate the mechanics of the library:
	A command module implements the command:

defmodule Demo.HelloCmd do
  use Tux.Command

  @impl true
  def about(), do: "Greet current user"

  @impl true
  def main(env, args), do: {:ok, "Hello #{env.pre.user}!"}
end
	A dispatcher module associates command names with command modules,
or groups them as sub-commands:

defmodule Demo do
  use Tux.Dispatcher

  # Command registration (with options)
  cmd "hello", Demo.HelloCmd, preloads: [:user]

  # A preload function can run prior to command execution
  def user(env), do: System.fetch_env!("USER")
end
	Build your elixir app as an escript executable. Note that you need to
update your mix.exs to reflect your escript's main module.
See mix escript.build
for more details.

# mix.exs
def project do
  [
    escript: [main_module: Demo]
    ...
  ]
end
Generate the escript executable:
$ mix escript.build
Generated escript demo

$ ./demo hello
Hello tuxuser!

	Extend your app with new commands simply by creating new
command modules, then register them with the dispatcher:

defmodule Demo.AddCmd do
  use Tux.Command

  @impl true
  def main(env, [x, y]) do
    {a, ""} = Integer.parse(x)
    {b, ""} = Integer.parse(x)
    {:ok, "#{a} + #{b} = #{a+b}"}
  end
end
Register the Demo.AddCmd module under the "add" name:
defmodule Demo do
  ...
  cmd "add", Demo.AddCmd
  cmd "hello", Demo.HelloCmd, preloads: [:user]
end
Rebuild the escript then run:
$ mix escript.build
$ ./demo add 1 2
1 + 2 = 3


  
    
  
  Overview


Tux was designed to be modular and dependency-free, all the while leveraging
the powerful Elixir macro system to enable the development of command line
interfaces with ease.

  
    
  
  0. Concepts


Here's a summary of some the concepts used throughout the tux universe:
	Command – some functionality implemented by an executable.
	Command Name – some keyword which identifies a command.
	Command Module – a specialized module which implements a given command.
	Command Dispatcher – a specialized module responsible for grouping commands
and for delegating program execution to the appropriate command module
in response to command invocations.
	Command Preloads – optional functions dispatchers can execute prior to
executing commands, and whose results are made available to the command modules
as part of their environment.


  
    
  
  1. Commands


Commands expose the functionality of your program/escript to the end user.
They are invoked using keywords (command names), as part of your
program/escript invocation.
By analogy, when you make use of the mix program and type mix compile,
you invoke the compile command of the mix program.
To implement commands using tux, you create command modules which
contain the business logic, then group them together under one or more
dispatchers.
Here's a the most minimal example of a command module created with tux:
defmodule PingCmd do
  use Tux.Command

  @impl true
  def main(env, args) do
    {:ok, "pong"}
  end
end
Command modules can implement other callbacks besides the Tux.Command.main/2.
See the Tux.Command behaviour for more details.
Alright, so now that we have just implemented a command, how can we connect it
to the rest of the escript? Enter dispatchers:

  
    
  
  2. Dispatchers


Dispatcher Module
   ├─── Command Module
   ├─── Command Module
   └─── Dispatcher Module
            ├─── Command Module
            └─── Command Module
Dispatchers are modules which contain the mappings command names -> command modules
and are responsible for delegating execution to the appropriate command module
in response to a command invocation.
Well, they actually do a little more than that, among which collecting the results
from command preloads, setting up the command context, recovering
from exceptions and more.
A dispatcher's functionality is implemented by the Tux.Dispatcher module, and it can
be injected into your own dispatcher modules via the __using__ macro.
Here's an example of a dispatcher which associates the ping command
with the PingCmd module we've just implemented:
defmodule MyProgram do
  use Tux.Dispatcher

  # Register a new command module via a command name
  cmd "ping", PingCmd

  # Alternatively a command module can be registered under multiple
  # names at once (useful for command shortcuts)
  cmd ~w"pong p", PongCmd
end
Customizing Dispatchers – Please read the section
Dispatcher Options for how to
customize a dispatcher's behaviour.

  
    
  
  3. Dispatcher & Command Preloads


Oftentimes, commands need to obtain some data or run some business logic prior
to the actual command execution, so to help with these aspects tux offers preloads.
Preloads are functions a dispatcher can execute prior to command execution,
and whose results can be made available to the command module in the env struct,
which is the first argument pass to the command module's Tux.Command.main/2
callback.
For illustrative purposes, let's update our ping command with a preload which
can return the current datetime and make that available in the command output:
Here's the updated dispatcher module and command registration:
defmodule MyProgram do
  use Tux.Dispatcher

  @doc "A preload function to return the current datetime"
  def date(_env), do: DateTime.utc_now()

  # Command registered with preload
  cmd "ping", PingModule, preloads: [:date]
end
And here's the command module making use of the preload's result:
defmodule PingCmd do
  use Tux.Command

  @impl true
  def main(env, args) do
    {:ok, "pong #{date}"}
  end
end
$ mix escript.build
$ ./program ping
pong 2024-06-07 20:36:56.687017Z


  
    
  
  4. Results


Command modules should return the following results:
	:ok
	:error
	{:ok, String.Chars.t()}
	{:error, String.Chars.t() | Tux.Alertable.t()}


Please sees String.Chars and Tux.Alertable protocols for more details.

  
    
  
  5. Errors


To signal a command has failed you can return a Tux.Result error tuple containing a
a Tux.Error (or a string message) value. In such a case, this value will be converted
and displayed nicely as a Tux.Alert.
Now, if the business logic of a command is deeply nested, it might be easier to
simply raise an exception from the depths of your code, with the downside of having
the end user see the entire stacktrace (which might not be an issue for
other developers), but nonetheless it can be unsightly.
With tux, you can rescue from command exceptions and display alerts instead. To this
end, you need to construct the dispatcher module with the rescue: true option, and
as a consequence, command exceptions will be shown more elegantly while still
preserving the ability to view the entire stacktrace when the command is invoked
with the --debug flag.
Exit Codes – A Tux.Error also contains an exitcode field, which you can
overwrite, and it will be used as the program/escript's exit status.

  
    
  
  6. Help


Tux includes the Tux.Help module to assist you in creating well-structured
command help message.
When a command implements the help/0 callback, if the said command
is invoked with the -h or --help flags, it will show something like this:
ABOUT
  scan - scan strings

USAGE
  strings scan [OPTS] [ARGS]

OPTIONS
  --uppercase, u    Uppercase endpoint name
  --lowercase, u    Lowercase endpoint name

NOTES
  Here are some additional notes to include
  in the help message.

Here's the pipeline used to create the above help message:
defmodule ScanCmd do
  use Tux.Command

  @impl true
  def about(), do: "Manage systems"

  @impl true
  def main(_, _), do: {:ok, "System managed"}

  @impl true
  def help() do
    Help.new()
    |> Help.about("scan", "scan strings")
    |> Help.usage(["strings scan [OPTS] [ARGS]"])
    |> Help.options([
      {"--uppercase, u", "Uppercase target name"},
      {"--lowercase, u", "Lowercase target name"}
    ])
    |> Help.section("notes", """
    Here are some additional notes to include
    in the help message.
    """)
    |> Help.ok()
  end
end

  
    
  
  7. Internals


For quick reference, here's an overview of the most important modules
of the tux library:
	Tux.Alert      - module for displaying errors (stylized)
	Tux.Command    - command module behaviour
	Tux.Config     - functions for reading and writing text based config files.
	Tux.Dispatcher - macros for command creation and dispatch
	Tux.Env        - a struct to encode preload result and other info
	Tux.Error      - functions user friendly errors construction.
	Tux.Exec       - execute a command module
	Tux.Help       - functions for creating help messages.
	Tux.Init       - construct env prior to command execution
	Tux.Prompt     - functions for prompting and type conversion.
	Tux.Result     - the type returned by commands.
	Tux.Show       - functions for user-friendly output.


More Examples – A series of runnable examples illustrating
the usage of the tux library can be found in the
examples folder
of the source repository.

      




  

    
Tux.Command behaviour
    



      
A command module provides the implementation for a given command,
and it can be registered on a given dispatcher using the
Tux.Dispatcher.cmd/2 macro.
At a minimum, a command module needs to implement only the main/2 callback.
The full list of callbacks is as follows:
	main/2 (required)
– implements the command logic.
	about/0 (optional)
– returns the command's short description.
	help/0 (optional)
– returns command's help message – see Tux.Help.
	show/2 (optional)
– writes final command output.
	exit/2 (optional)
– stop the VM with an appropriate exit code.


  
    
  
  Example


Here's an example of a typical command module:
defmodule ExampleCmd do
  use Tux.Command

  @impl true
  def about(), do: "this command does something"

  @impl true
  def main(env, args), do: {:ok, "some result"}

  @impl true
  def help() do
    Help.new()
    |> Help.about(about())
    |> ...
    |> Help.ok()
  end
end

Tux.Dispatcher behaviour
    



      
Dispatchers are modules which control program flow by directing the command execution
to the appropriate command module in response to a given command/sub-command
invoked by the CLI's end user.

  
    
  
  Creating Dispatchers


New dispatchers are created by invoking use Tux.Dispatcher in your own
modules, and then employing the various macros injected to register
Tux.Command modules for their respective command names.

  
    
  
  Example


defmodule MyProgram do
  use Tux.Dispatcher, opts

  cmd "ping", PingCmd, preloads: [:auth]
  cmd "pong", PongCmd, preloads: [:auth, :log]

  cmd "group", AnotherDispatcherWithSubcommands

  cmd ~w"h hello", HelloCmd
  cmd ~p".a", CommandTriggeredByDotAPrefix

  # Alternatively
  pre [:config, :auth, :log] do
    cmd "ping", PingCmd
    cmd "pong", PongCmd
  end
end

  
    
  
  Dispatcher Options


A dispatcher's behaviour can be customized by providing a list of options:
use Tux.Dispatcher, [... rescue: true, colors: false, ...]
Available dispatcher options are:
	:rescue - a boolean flag to rescue from command exceptions (default is false)
	:device - the IO.device/0 to write the command output to
(default is :stdio)
	:newlines - a boolean flag to append new lines to outputs for nicer
outputs (default is true)
	:colors - a boolean flag to enable/disable colored outputs (default is true)
	:preloads - a list of functions to execute prior to executing a command module
(default is []). The results of these preloads will be accessible
in the command module's Tux.Command.main/2 callback as env.pre.*,
where * is the function name or the key (if one was specified).
– See Dispatcher Preloads below for more details.
	:exitwith - the name of the function of the System module to invoke
in order to terminate the VM when a command returns an error result.
Accepted values are :halt and :stop. (default is :halt).
See Erlang VM Exits below for more details.


  
    
  
  Erlang VM Exits


When a command module returns an error, tux will terminate your program with
a non-zero exit code, so the failure can be reflected in the shell status code.
By default, in the case of errors tux terminates the program with System.halt/1
(the default escript behaviour) because it is faster and we want responsive CLIs.
However, if for whatever reason you need your system to gracefully come to a stop
(say you have processes which need to perform come cleanup), you can change
this behaviour by specifying exitwith: :stop in the dispatcher's options.

  
    
  
  Dispatcher Preloads


In many cases, prior to the actual execution of a command, some piece of data
is needed (e.g. data read from a config file, etc). And since this pattern could
recur across many commands of a given CLI, tux introduces the concept of
preloads.
Preloads are functions to be executed sequentially, prior to the invocation of
a command module, and whose results will be collected under the .pre key of the
Tux.Env struct passed to the command module's Tux.Command.main/2 callback.
Preloads can be declared at both the dispatcher definition and
command registration levels, and are executed sequentially in the order in which
they were defined in the tux command hierarchy. Thus, the can be declared:
	At dispatcher level – using the :preloads dispatcher option. (executed 1st)
	At command registration level - using the pre/2 (2nd) and cmd/3 (3rd) macros.


  
    
  
  Preloads Format


Preloads can specified using any of the following formats (Note that key, fun,
and mod are atoms) and are found under the Tux.Envs .pre key:
	Preload Format	Example	Preload Result
	fun	:config	env.pre.<fun>
	{key, fun}	{c!: :config}	env.pre.<key>
	{mod, fun, args}	{MyPre, :config, []}	env.pre.<fun>
	{key, {mod, fun, args}}	{:mc!, {MyPre, :config, []}}	env.pre.<key>

When a preload is specified using a function name, that function must exist
in the dispatcher where that preload is being specified.

  
    
  
  Preloads Examples


Here's a list of preloads specified using multiple formats at once:
defmodule MyPreloads do
  def config(env), do: "My config file"
  def config(env, :upcase), do: "MY CONFIG FILE"
end

defmodule MyDispatcher do
  use Tux.Dispatcher

  cmd "show", MyCmd, preloads: [:conf,
                                c1: :conf,
                                c2: {MyPreloads, :config, []},
                                c3: {MyPreloads, :config, [:upcase]}
                               ]

  def conf(env), do: "My config file"
end

defmodule MyCmd do
  use Tux.Command

  def main(env, _) do
    IO.inspect(env.pre.conf)
    IO.inspect(env.pre.c1)
    IO.inspect(env.pre.c2)
    IO.inspect(env.pre.c3)
  end
end
And here's an example with preloads specified at multiple levels
in the tux command hierarchy:
defmodule MyDispatcher do
  use Tux.Dispatcher, preloads: [:auth, one: :inc]

  def auth(_env), do: true
  def inc(_env), do: 1
  def inc(env, prev), do: env.pre[prev] + 1

  defmodule CollectCmd do
    use Tux.Command

    @impl true
    def main(env, _args) do
      true = env.pre.auth
      {:ok, "Summing all preloads should yield three: #{env.pre.three}"}
    end
  end

  pre [two: {MyDispatcher, :inc, [:one]}] do
    cmd "collect", CollectCmd, preloads: [three: {MyDispatcher, :inc, [:two]}]
  end
end

  
    
  
  Dispatcher Internals


A dispatcher is broken up internally into various stages:
	Tux.Init - for command preloads execution and Tux.Env initialization
	Tux.Exec - for command execution
	Tux.Show - for command output display
	Tux.Exit - for command exits & stopping the VM
	Tux.Help - for dispatcher help message generation

Tux.Case 
    



      
Utility macros to generate test cases for dispatchers or main
escript modules.

  
    
  
  Examples


The following example uses the scenario/2 macro to create the
entire ExUnit.Case test cases in a single step:
defmodule MyTests do
  use Tux.Case

  scenario "check for EXACT output",
    using: SomeDispatcher,
    invoke: "ping --flag --flag",
    expect: [exactly: "pong\n"]

  scenario "check for APPROXIMATE command output",
    using: SomeDispatcher,
    invoke: "ping --flag --flag",
    expect: [approx: "pong"]

  scenario "check command output using CUSTOM FUN",
    using: SomeMainModule,
    invoke: "ping --flag --flag",
    expect: fn output ->
      assert String.contains?(output, "pong")
    end
end
This module also includes the execute/2 macro, which does not setup a test case
and only asserts on the command output. This is useful if you need control
over how you create your test cases:
defmodule MyTests do
  use Tux.Case

  test "some test using the lower level execute macro" do
    execute SomeDispatcher,
      invoke: "ping --with --some --flags"
      expect: [exactly: "pong\n"]
  end
end

Tux.Colors 
    



      
Basic functions wrapping a few IO.ANSI coloring functions
for writing stylized output to the terminal.
bold/2 underline/2 + faint/2 gray/2
red/2 orange/2 yellow/2 green/2 blue/2
When you use Tux.Command this module is imported automatically
along with the Tux.Help module.

  
    
  
  Usage


import Tux.Colors

"""
This is some #{bold("bold text")} illustrating the #{underline("usage")}
of Tux.Colors module. This #{red("word", false)} however,
won't be colored because the second argument passed is false, which is useful
when you want to control the colored output based on some configuration flag
passed to the functions in this module.
"""

Tux.Config 
    



      
Basic support for reading and writing key = value configuration files,
with support for ignoring comment lines (lines that start with #
or ;).

  
    
  
  Example


Say you have an existing configuration file file.conf:
# This is a comment line (will be ignored)
path = /tmp

# Note that all values will be parsed as strings
rounds = 10

; Windows-style comments are also ignored,
; and so are these [sections]
[couting]
count = 123

You can read it with read_file/2:
with {:ok, config} <- Tux.Config.read_file("file.conf") do
  assert config["path"] == "/tmp"
  assert config["rounds"] == "10"
  assert config["count"] == "123"
end

  
    
  
  Notes


	All read values will be strings.
	All read valeus which are empty strings will be converted to nil.

Tux.Env 
    



      
The environment struct contains the information required by a
dispatcher to execute the logic associated with a given CLI command,
while also being the first argument to be passed to a command module's
Tux.Command.main callback.

  
    
  
  Struct Fields


	:raw - the list of all escript args as typed by the user
	:cmd - the command name invoked by the end user which matched
against dispatcher's registered command names.
(Note that inside a dispatcher, command registration can be achieved
using multiple variants: a keyword, a list of keywords, a prefix. – See
Tux.Dispatcher.cmd/2 for more details).
	:mod - the command module associated with the given command name.
	:fun - the command module callback to execute – .e.g. :main or :help.
This appropriate function is computed based on the given CLI arguments.
For example, when a -h or --help flag was provided, the dispatcher will
invoke the command module's :help callback, otherwise the :main.
	:arg - the CLI arguments (minus the command name) to pass to the command module
for command execution.
	:dev - the IO device where output will be written. (See Tux.Show).
	:pre - the map with the results from all registered preloads
	:dsp - the context/parent dispatcher module.
	:new - the boolean flag to add newlines to shown success results.
	:ext - the System function to invoke when stopping the VM runtime.
The accepted values are :halt or :stop.

Construct custom exception modules which automatically derive the
Tux.Alertable protocol. The generated error type will have the
following fields:
	:message   - short error message which will be used as title
	:details   - detailed explanation of the error (.e.g. info about possible
resolutions, etc)
	:exitcode - exit code to be used at escript's termination


  
    
  
  Use Options


When constructing new error modules with use Tux.Error, the following
options are available:
	:message (required) – main error message to use for the struct
	:exitcode (optional) – exit code (default is 1)


  
    
  
  Examples


defmodule MyError do
  use Tux.Error, message: "MyError Occurred"
end

defmodule MyOtherError do
  use Tux.Error, message: "MyError Occurred", exitcode: 100
end
Now you can raise or return the error from your command modules:
raise MyError
{:error, %MyError{...}}
A useful idea is to define a new/1 function in your error module
which accepts specific information and constructs a more detailed error:
defmodule FieldError do
  use Tux.Error, message: "Field is Missing"

  def new(fieldname: fieldname) do
    struct!(__MODULE__, details: """
    The #{red(fieldname)} is missing.
    """)
  end
end
Here's how the Tux.Alertable protocol behaves when
invoked with a tux error struct:
defmodule MyError do
  use Tux.Error, message: "Some Error Occurred"
end

error = %MyError{}
assert Tux.Alertable.title(error) == "Some Error Occurred"
assert Tux.Alertable.message(error) == nil

error = MyError{details: "This error is really, really bad."}
assert Tux.Alertable.title(error) == "Some Error Occurred"
assert Tux.Alertable.message(error) == "This error is really, really bad."

Create beautiful looking, well-structured help messages
for commands or dispatchers.

  
    
  
  Features


	Elegant layouts
	Minimal coloring
	Customizable


  
    
  
  Example


Here's an example of a well structured command help message
which you can create with tux:
ABOUT
  mng - command for controlling systems

USAGE
  tool mng [OPTS] [ARGS]

OPTIONS
  --upcase, u        Uppercase endpoint name
  --lowercase, l     Lowercase endpoint name

NOTES
  Here are some additional notes to include
  in the help message.


  
    
  
  Pipelined Help Messages


The help message above can be created with the following Tux.Help pipeline:
defmodule MngCmd do
  use Tux.Command

  @impl true
  def main(_, _), do: ...

  @impl true
  def help() do
    Help.new()
    |> Help.about("mng", "command for controlling systems")
    |> Help.usage(["tool mng [OPTS] [ARGS]"])
    |> Help.options([
      {"--upcase, u", "Uppercase endpoint name"},
      {"--lowercase, l", "Lowercase endpoint name"},
    ])
    |> Help.section("notes", """
    Here are some additional notes to include
    in the help message.
    """)
    |> Help.ok()
  end
end

  
    
  
  Manual Help Messages


If you want complete control over your help messages, you can construct
them manually. Here's the code that would render the above help message:
defmodule MngCmd do
  use Tux.Command

  @impl true
  def help() do
    """
    #{bold("ABOUT")}
      mng - command for controlling systems
    #{bold("\nUSAGE")}
      #{green("tool mng [OPTS] [ARGS]")}
    #{bold("\nOPTIONS")}
      #{green("--upcase, u")}        Uppercase endpoint name
      #{green("--lowercase, l")}     Lowercase endpoint name
    #{bold("\nNOTES")}
      Here are some additional notes to include
      in the help message.
    end
    """
    |> Help.ok()
  end
  end

Tux.Prompt 
    



      
Helper functions for retrieving user input as: strings, integers, floats
or for asking for confirmations.
	Default values will be displayed between [...] at the end of the prompt.
	Supports default values when no value (just enter) is typed.


  
    
  
  Examples


alias Tux.Prompt

with true <- Prompt.for_confirmation("Really continue"),
     {:ok, name} <- Prompt.for_string("What's your name", "Joe Doe"),
     {:ok, age} <- Prompt.for_string("What's your age"),
     {:ok, salary} <- Prompt.for_float("What's your salary") do
      IO.puts("Name: #{name}, Age: #{age}, Salary: #{salary}")
end

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Tux.Result - Tux v0.4.0
    
    

    


  
  

    
Tux.Result 
    



      
The tux result is the standard result type which must be returned by a command module,
which controls what the CLI's end-user sees as output.
	Result Value	Meaning	Command Output Behaviour
	:ok	Success	nothing will be shown
	:error	Failure	a Tux.Alert will be shown
	{:ok, term}	Success	term will be shown (term must implement String.Chars).
	{:error, term}	Failure	a Tux.Alert (red)    will be shown (term must implement Tux.Alertable)


  
    
  
  Non-standard Results


You can also return anything else you like from a command, in which case you need
to implement your own show/2 callback in your command module to instruct tux
how to print your command's results.
There are some caveats to returning non-standard command results (see below),
however, case need be, here's an example of how you could do it:
defmodule Cli do
  use Tux.Dispatcher

  defmodule AddCmd do
    @impl true
    def about(), do: "Add a list of numbers"

    @impl true
    def main(_env, args) do
      args
      |> Enum.map(&parse/1)
      |> Enum.reject(fn value -> value == :error end)
      |> Enum.sum()
    end

    defp parse(string) do
      case Integer.parse(string) do
        {num, ""} -> num
        _ -> :error
      end
    end

    @impl true
    def show(env, result) do
      case "--json" in env.raw do
        true ->
            %{result: result}
            |> to_json_string()
            |> IO.puts()
        false ->
            IO.puts(result)
      end
    end
  end

  cmd "add", AddCmd
end

defmodule Math do
  def main(args), do: Cli.main(args)
end
$ ./math add 1 2 3 4
{"result": 10}


  
    
  
  Caveats for Non-standard Results


Tux determines your escript's exit code based on any given command's return value
if it follows the Tux.Result format, otherwise the exit code will always be 0
– see Tux.Exit – which might not be what you want if you want to signal a program error.
Of course, along with your custom show/2 implementation you can also implement
a custom exit/2 which checks the result and stops the VM with an appropriate code.
Here's an implementation extending the above example, which for fun
sets the escript's exit code to the computed sum value:
defmodule AddCmd do
  use Tux.Command

  ...

  @impl true
  def exit(env, result) do
    System.halt(result)
    # or
    # apply(System, env.ext, [result])
  end
end

Tux.Alert 
    



      
An alert is a struct which encodes the textual and formatting information
which can be used to display stylized error messages.

  
    
  
  Examples


Please be aware that the colored and embolden parts cannot be seen in this example,
nonetheless an alert in its simplest version looks roughly something like this:
│ ERR: Command error

And with additional details:
│ ERR: Command error
│
│ This section has more details about the error.

Alertable protocol is used by Tux.Show to display command error results
in a user-friendly fashion using the Tux.Alert struct.
The protocol is implemented for the following types:
	Strings – alertable is implemented for strings, so that if you return
{:error, "some message"} from a command, the message will be shown as
a Tux.Alert.

	Tux.Error - alertable is also implemented for the Tux.Error struct
or any struct which is constructed with use Tux.Error, so that if you return
{:error, %Tux.Error{}} or {:error, %MyDerivedError{}} from a command,
the struct will be shown as a Tux.Alert.

	Exceptions - alertable is implemented for the Elixir Exception structs
and will extract the :message field as the alert's title. Thus, if your app
throws an exception and the dispatcher is using the options rescue: true,
the thrown exception will be shown as an alert.

Tux.Errors 
    



      
Errors raised or returned by various tux components.

  
    
  
  Non-explainable Errors


	ConfigReadError
	ConfigWriteError
	NotImplementedError


  
    
  
  Alertable Errors


	CommandRescuedError
	CommandNotFoundError

Tux.Exec behaviour
    



      
Dispatcher subystem for invoking command modules, with optional rescuing
from raised exceptions.
This module is injected in dispatcher at call site:
...
use Tux.Exec, overrideable: true
...
Of course, the injected exec/2 function can be overwritten in
your dispatchers:
defmodule MyDispatcher do
  use Tux.Dispatcher

  @impl true
  def exec(env, exec_opts) do
    ... implement custom command module execution here
  end
end

Tux.Exit behaviour
    



      
Routines for stopping the VM with an appropriate exit code
from command modules.
This logic in this module is automatically injected by dispatcher and
command modules, however its callbacks can be overwritten for custom needs:
defmodule MyCmd do
  use Tux.Command

  @impl true
  def exit(env, result), do: ...stop the VM...
end

Tux.Init behaviour
    



      
Dispatcher subsystem used internally for initialization of command environment
and collection of preload results prior to command execution.

Find the command module for a given comm